Visual Basic

CHAPTER 7

實驗數據分析

(Data Analysis)

張 榮 興 博士

豐映科技股份有限公司

E-mail: chang.ronhsin@msa.hinet.net

http://www.resi.com.tw/vb.htm

研究人員窮畢生精力進行各種實驗工作

實驗數據不斷累積以驗證理論及建立經驗方程式

設計問題 D-VII

管內的強制對流熱傳

根據因次分析，管內的強制對流熱傳係數可表示成以下的函數關係：

$$Nu = F\left(Re\ ,\ Pr\ ,\ \frac{\mu_b}{\mu_a}\ ,\ \frac{L}{D}\right)$$

其中 $Nu = \dfrac{hD}{k}$ ，$Re = \dfrac{DU\rho}{\mu_b}$ ，$Pr = \dfrac{\rho C_P \mu_b}{k}$ ，μ_b 及 μ_a 分別為流體在平均溫度及平均管壁溫度時的黏滯度 [1]。

試根據下表所示的實驗數據，建立適當的經驗方程式。

L/D	Re	Pr	μ_b/μ_a	Nu
82	200	490	1.02	20.09
	560	373	2.11	28.82
	1000	330	1.72	31.47
	1800	315	1.74	38.11
	2100	280	2.12	40.58
97	200	0.71	1.00	3.65∗
	400	0.71	1.00	3.68∗
	700	0.71	1.00	3.67∗
	900	0.71	1.00	3.47
	1200	0.71	1.00	3.76
	1560	0.71	1.00	4.20
	1710	0.71	1.00	4.25
	2000	0.71	1.00	4.53
220	600	7000	2.09	54.20
	750	7200	2.21	59.39
	870	5400	1.98	55.72
	930	5600	1.97	58.09
	1620	1730	1.98	47.60
	1900	1650	1.99	49.07

註：表中有∗註記的三點數據若不納入，結果有何變化？

工程科學是一門由經驗累積，並且相當仰賴實驗數據的科學，因此，有非常多的研究人員窮畢生精力進行各種領域的實驗工作，各種實驗數據也不斷地累積，以驗證理論及建立經驗方程式。而在實際應用時，零亂未經整理的數據，在應用上相當困難，並且沒有價值；有效率且規律性的實驗數據整理與處理，對工程及科學而言是相當重要的。實驗數據的整理與處理，通常必須兼顧基本原理的正確性及使用時的正確性與方便性。一般而言，實驗數據資料處理通常使用以下三種方法：

1. 表格表示法：例如空氣比熱 C_p 與溫度及壓力的關係，如表 7.1 所示；希望求得某一溫度、壓力時的 C_p 值，可利用本書第三章所介紹的插值法求之。

表 7.1　空氣之比熱 C_p　(J/mole-K)

T(K)	*P*(atm)			
	1	10	20	30
180	29.52	31.24	33.51	36.14
200	29.38	30.59	32.10	33.73
220	29.26	30.14	31.23	32.38
240	29.16	29.88	30.72	31.60

2. 圖形表示法：將函數關係製作成圖形，使用時再按圖索驥，由圖讀取函數值。這種方法最大的缺點是讀數據時準確度較差。
3. 函數近似法：將實驗所得數據利用一經驗方程式表示。經驗方程式的建立，通常需仰賴圖形表示法先建立大略的函數關係，或佐以理論基礎，再判斷可使用的函數型態，然後以數學方法建立經驗式。

本章將利用實際數據的處理來介紹 (2) 及 (3) 兩種使用方法。

第一節　圖形分析法

Visual Basic

實驗所得原始數據，有時看似雜亂無章，但如果先作適當處理，然後再作圖，則很可能使原先看似不規則的實驗數據，顯現出規則的函數關係。這種實驗數據的處理，通常需建立在對該一現象或觀察的理論分析上；但對於一項嶄新的實驗觀察，有時也可能由實驗結果分析及各種歸納，引導出新理論的建立。

以表 7.2 的數據爲例，直接作圖，可以得到如圖 7.1 所示之曲線關係。

表 7.2　典型的數據表

x	y	x/y
0.00	0.00	-
0.50	1.00	0.50
1.21	1.50	0.81
3.10	1.92	1.61
4.17	2.01	2.07
4.98	2.05	2.43
5.77	2.09	2.76
9.00	2.17	4.15
12.10	2.21	5.48
17.00	2.24	7.59



圖 7.1　y 對 x 作圖

但若先計算 x/y 值，再以 (x/y) 對 x 作圖，則可以得到如圖 7.2 所示的直線關係。以方程式表示之，得到

$$\frac{x}{y} = m\,x + b = \frac{3}{7}x + \frac{2}{7} \tag{7-1.1}$$



圖 7.2　x/y 對 x 作圖，成一直線關係

其中 *m* 爲圖 7.2 所示直線的斜率，b 爲截距。移項整理後，得到 y 與 x 的函數關係爲

$$y = \frac{7x}{2+3x} \tag{7-1.2}$$

工程上常用的作圖格式，包括正交作圖、半對數作圖、全對數作圖及或然率作圖等，分別有其使用上的特色與用途。以下分別就各種函數型式的適當作圖方法作一簡要說明。

$y = m x^n$

要將這類函數轉換成直線關係，可對原方程式左右同時取對數，得到

$$(Log\ y) = (Log\ m) + n\,(Log\ x) \tag{7-1.3}$$

此時，若令 Y=$Log\ y$, X=$Log\ x$，則上式可改寫成

$$Y = a + nX \tag{7-1.4}$$

爲一直線方程式，如圖 7.3 所示，利用全對數座標製作 Y 對 X 的圖形，可以得到直線關係。



圖 7.3　函數 $y = m\,x^n$，$m = 1$

工程上常用對數座標紙，將數據直接繪製於圖紙上，若所得結果呈現類似圖 7.3 的線性關係，則可推斷原始數據可利用函數關係 $y = m\,x^n$ 表示。

$y = e^{mx}$

這類型的函數曲線如圖 7.4 所示，其特點為：

1. 一定通過 (0 , 1) 這一點。
2. 若 $m > 0$，則當 x 增大，y 會趨近於無窮大；但若 $m < 0$，則當 x 增大，y 會趨近於 0。

要將此方程式轉換成直線關係，同樣可對原方程式取對數，得

$$(\log y) = mx \tag{7-1.5}$$

令 $Y = \log y$，$X = x$，則 Y 對 X 作圖為一通過原點的直線。圖 7.5 即函數 $y = e^{mx}$ 利用半對數紙作圖所得結果，原方程式可被轉換成線性關係。



圖 7.4　$y = e^{mx}$



圖 7.5　函數 $y = e^{mx}$

$y=\dfrac{cx}{1+cx}$

此類型的函數曲線如圖 7.6 所示，其特點為：

1. 一定通過原點 (0 , 0)。
2. 當 x 增大時，y 會趨近於 1。
3. c 值愈大，y 趨近於極限值 1 的速率即愈快。

這類函數常見於反應速率表示式。轉換成線性關係式的方法，可利用下列二種方式：

1. 若將函數 $y=\dfrac{cx}{1+cx}$ 作適當轉換，令 $Y=1/y$，$X=1/x$，則原方程式可改寫成

$$Y=1+\frac{1}{c}X \qquad \textbf{(7-1.6)}$$

所得到直線圖形如圖 7.7 所示。當 x 趨近無限大時，$1/x$ 趨近於零，y 值趨近於 1。當參數 c 趨近於無限大時，x 值變化之影響即變得不明顯，y 值趨近於 1。



圖 7.6 函數 $y=\dfrac{cx}{1+cx}$

2. 若將函數 $y=\dfrac{cx}{1+cx}$ 作適當轉換，令 $Y=x/y$，$X=x$，則原方程式可改寫成：

$$Y=\frac{1}{c}+X \qquad \textbf{(7-1.7)}$$

直線圖形如圖 7.8 所示，斜線均為 1，截距則隨 c 值而變。c 值趨近於無窮大時，所得直線位於圖形對角線上，即 $Y=X$。



圖 7.7　函數 $y=\dfrac{cx}{1+cx}$ 之轉換



圖 7.8　函數 $y=\dfrac{cx}{1+cx}$ 之轉換

例題 7-1　氣體狀態方程式係數推定

已知苯蒸氣在 290°C 時的莫耳體積如表 7.3 所示。假設苯蒸氣的 *P-V-T* 關係可利用維瑞耳狀態方程式 (Virial Equation of State) 表示：

$$P = \frac{RT}{V}\left(1 + \frac{\beta}{V} + \frac{\gamma}{V^2}\right) \tag{7-1.8}$$

試求係數 β 及 γ 值。

表 **7.3**　苯蒸氣在 290°C 時的莫耳體積

P (atm)	V (cm^3/mole)	P (atm)	V (cm^3/mole)
30.64	1,114	40.04	707
31.60	1,067	41.79	646
32.60	1,013	43.59	591
33.89	956	45.48	506
35.17	900	47.07	443
36.63	842	48.07	386
38.39	771		



圖 **7.9**　苯蒸氣的莫耳體積與壓力關係圖

解：

將原始數據 (P，V) 作圖，得一略呈 S 型的曲線，如圖 7.9 所示。尤其當壓力 P 增大時，其線性關係有較明顯變化。

維瑞耳狀態方程式可改寫成線性函數：

$$〔V(\frac{PV}{RT}-1)〕=\beta+\gamma〔\frac{1}{V}〕 \tag{7-1.9}$$

令 $Y=〔V(\frac{PV}{RT}-1)〕$，$X=\frac{1}{V}$，則原方程式可改寫成

$$Y=\beta+\gamma X \tag{7-1.10}$$

為一直線，由其斜率及截距即可分別求得 β 及 γ。

圖 7.10 即為 $〔V(\frac{PV}{RT}-1)〕$ 對 [1/V] 作圖所得結果，由圖得斜率 $\gamma=3.53\times10^4$，截距 $\beta=-323$。



圖 **7.10**　維瑞耳狀態方程式之線性化

第二節　最小平方法

Visual Basic

前一節簡單的介紹了幾種函數的線性化方法，可是如例 7-1 所示，數據利用適當轉換後，雖然可以變成直線關係，但要如何在所繪製的數據點之間劃一條最具代表性的直線呢？「最佳近似」直線的定義有許多種不同的說法，其中最常見也被使用得最廣泛的，就是能*使所有的數據點與近似方程式誤差的平方和爲最小的直線*，這種方法即稱爲「最小平方法」(Least Square Method)。

假設實驗所得數據爲 (x_i , y_i)，$i = 1 , 2 , ... , n$。令所有數據點的「最佳近似」直線爲

$$\hat{y} = \alpha + \beta x \qquad \textbf{(7-2.1)}$$

此直線與 n 個數據點誤差（ 近似値與實側値之差 ）之平方和爲：

$$S = \sum_{i=1}^{n} [(y_i - \hat{y}_i)^2] = \sum_{i=1}^{n} [(y_i - \alpha - \beta x_i)^2] \qquad \textbf{(7-2.2)}$$

要求得最佳係數 α 及 β，使誤差的平方和 S 變成最小，則需滿足以下二條件:

$$\frac{\partial S}{\partial \alpha} = 0 \quad ; \quad 即 \quad -2\sum_{i=1}^{n}(y_i - \alpha - \beta x_i) = 0 \qquad \textbf{(7-2.3)}$$

$$\frac{\partial S}{\partial \beta} = 0 \quad ; \quad 即 \quad -2\sum_{i=1}^{n}[x_i \ (y_i - \alpha - \beta x_i) = 0 \qquad \textbf{(7-2.4)}$$

方程式 (7-2.3) 及方程式 (7-2.4) 展開後，可分別改寫成

$$(n)\ \alpha + (\sum_{i=1}^{n} x_i)\ \beta = (\sum_{i=1}^{n} y_i) \qquad \textbf{(7-2.5)}$$

$$(\sum_{i=1}^{n} x_i)\ \alpha + (\sum_{i=1}^{n} x_i^2)\ \beta = (\sum_{i=1}^{n} x_i\ y_i) \qquad \textbf{(7-2.6)}$$

得到二聯立方程式，求解兩個未知變數 α 及 β。解以上二聯立方程式，得到 α 及 β 分別爲：

$$\beta = \frac{n\,(\sum x_i y_i) - (\sum x_i)(\sum y_i)}{n\,(\sum x_i^2) - (\sum x_i)^2} \qquad \textbf{(7-2.7)}$$

$$\alpha = \frac{(\sum y_i) - \beta(\sum x_i)}{n} \qquad \textbf{(7-2.8)}$$

例題 7-2　最小平方法

利用本節所介紹的最小平方法，試求例 7.1 中的係數 β 及 γ 值。

解：維里狀態方程式可改寫成線性函數：

$$〔V(\frac{PV}{RT}-1)〕= \beta + \gamma〔\frac{1}{V}〕 \qquad \textbf{(7-2.9)}$$

令 $Y =〔V(\frac{PV}{RT}-1)〕$，$X = \frac{1}{V}$，則原始數據可整理如表 7.4。

然後分別求得 $\sum X_i$ ，$\sum Y_i$ ，$\sum X_i^{\,2}$ ，$\sum X_i Y_i$ 爲

$$\sum X_i = 18.942 \times 10^{-3}$$

$$\sum Y_i = -3527.1$$

$$\sum X_i^{\,2} = 31.051 \times 10^{-6}$$

$$\sum X_i Y_i = -5.0176$$

表 7.4　苯蒸氣在 290°C 時的莫耳體積

p(atm)	V(cm^3/mole)	$X = 1/V$	$Y =〔V(\frac{PV}{RT}-1)〕$
30.64	1,114	8.98E-04	−291.14
31.60	1,067	9.37E-04	−288.46
32.60	1,013	9.87E-04	−289.06
33.89	956	1.05E-03	−285.73
35.17	900	1.11E-03	−283.51
36.63	842	1.19E-03	−280.01
38.39	771	1.30E-03	−277.15
40.04	707	1.41E-03	−273.89
41.79	646	1.55E-03	−268.60
43.59	591	1.69E-03	−261.52
45.48	506	1.98E-03	−254.01
47.07	443	2.26E-03	−243.10
48.07	386	2.59E-03	−231.01

代入方程式 (7-2.7) 及方程式 (7-2.8)，即可得到

$$\beta = 35277.51 \quad ; \quad \alpha = -322.75$$

這種計算利用 Excel 撰寫程式最爲簡單，利用 Excel 所建立表格如表 7.5 所示，詳細程式編寫方法，可參考所附光碟。

表 **7.5** 利用 Excel 計算例 7.2 之結果

N	X	Y	X^2	XY
	1/V	V*(PV/RT-1)	(1/V)^2	(1/V)*(V*(PV/RT-1))
1	8.98E-04	–291.18	8.06E-07	–2.61E-01
2	9.37E-04	–288.50	8.78E-07	–2.70E-01
3	9.87E-04	–289.10	9.74E-07	–2.85E-01
4	1.05E-03	–285.76	1.09E-06	–2.99E-01
5	1.11E-03	–283.54	1.23E-06	–3.15E-01
6	1.19E-03	–280.04	1.41E-06	–3.33E-01
7	1.30E-03	–277.18	1.68E-06	–3.60E-01
8	1.41E-03	–273.91	2.00E-06	–3.87E-01
9	1.55E-03	–268.62	2.40E-06	–4.16E-01
10	1.69E-03	–261.54	2.86E-06	–4.43E-01
11	1.98E-03	–254.02	3.91E-06	–5.02E-01
12	2.26E-03	–243.11	5.10E-06	–5.49E-01
13	2.59E-03	–231.01	6.71E-06	–5.98E-01
SUM	1.89E-02	–3.53E+03	3.11E-05	–5.02E+00
β	= (N*SUM(XY)-(SUM(X)*SUM(Y))/((N*SUM(X^2)-(SUM(X))^2))			
	= 35,277.51			
γ	= (SUM(Y)-Beta*SUM(X))/N			
	= -322.7499694			

符號說明：

N = 數據點數

SumX = $\sum X_i$

SumY = $\sum Y_i$

SumX2 = $\sum X_i^2$

SumY2 = $\sum Y_i^2$

SumXY = $\sum X_i Y_i$

X,Y = 數據點

程式列印：

```
'    *******************************
'    LINEAR REGRESSION
'    *******************************
'
Private Sub LinearRegression(Xpos, Ypos)
Dim X(100), Y(100)
Cls
Print "ENTER DATA POINTS:"
N = 0
Do
        N = N + 1
        Print "Enter X("; N; ") = ";
        X(N) = Val(InputBox("Enter X Value = ", "X", X(N), Xpos, Ypos))
        Print X(N);

        Print "            Y("; N; ") = ";
        Y(N) = Val(InputBox("Enter Y Value = ", "Y", Y(N), Xpos, Ypos))
        Print Y(N)

        NextDataYN$ = InputBox("Next Data <Y/N>", "NEXT DATA", "Y", Xpos, Ypos)
Loop While NextDataYN$ = "Y"
'
' Convert to linear form by user
'
Call ConvertLinear(X, Y, N)

Call Regression(X, Y, Alpha, Beta, N)
MsgBox ("Ready for Data Interpolation")

Call LinInterpolation(Alpha, Beta)
End Sub


Private Sub ConvertLinear(X, Y, N)
'
'Example 7.2
'
For I = 1 To N
        HOLD = Y(I)
        Y(I) = Y(I) * (X(I) * Y(I) / 82.06 / 563.15 - 1)
        X(I) = 1 / HOLD
Next I
End Sub
```

```
Public Sub Regression(X, Y, Alpha, Beta, NoData)
'
' Linear Regression Subroutine
'
' X           = Independent Variable
' Y           = Dependent Variable
' Alpha       = Parameter
' Beta        = Parameter
' NoData      = Number of Data Points
'
' CLEAR MEMORY
'
SumX = 0
SumY = 0
SumX2 = 0
SumY2 = 0
SumXY = 0
'
' ===CALCULATE SUMS
'
For I = 1 To NoData
    SumX = SumX + X(I)
    SumY = SumY + Y(I)
    SumX2 = SumX2 + X(I) ^ 2
    SumY2 = SumY2 + Y(I) ^ 2
    SumXY = SumXY + X(I) * Y(I)
Next I
'
' ===Compute coefficients
'
Beta = (NoData * SumXY - SumY * SumX) / (NoData * SumX2 - SumX ^ 2)
Alpha = (SumY - Beta * SumX) / NoData

Print
Print "*** Linear Regression ***"
Print
Print "F(X) = ";
Print Format(Alpha, "0.000000E+00 ");
Print "+ (";
Print Format(Beta, " 0.000000E+00");
Print "* X )"
'
' ===Regression analysis
'
A = Beta * (SumXY - SumX * SumY / NoData)
B = SumY2 - SumY ^ 2 / NoData
```

```
C = B - A
D = A / B
Print "NUMBER OF OBSERVATIONS N = "; NoData
Print "Coefficient of determination (R^2) = ";
Print Format(D, "0.000000E+00")
Print "Coefficient of correlation = ";
Print Format(Sqr(D), "0.000000E+00")
Print "Standard error of estimate = ";
Print Format(Sqr(C / (NoData - 2)), "0.000000E+00")
End Sub


Public Sub LinInterpolation(Alpha, Beta)
'
' Linear Interpolation
'
' ===ESTIMATE Y-COORDINATES OF POINTS WITH ENTERED X-COORDINATES===
Cls
Print
Do
        Print "X = ";
        X = Val(InputBox("Enter X Value for Interpolation", "X", X, Xpos, Ypos))
        Print Format(X, " 0.000000E+00        ");
        Print "Y = ";
        Print Format(Alpha + Beta * X, "    0.000000E+00")
        NextDataYN$ = InputBox("Next Data <Y/N> ?", "Y/N", "N", Xpos, Ypos)
Loop While NextDataYN$ <> "N"
End Sub
```

副程式使用說明：

1. 副程式 **Private Sub ConvertLinear(X,Y,N)**

   輸入 X,Y,N，經線性化處理後，傳回線性化後之 X,Y 數據

2. 副程式 **Public Sub Regression(X,Y,Alpha,Beta,NoData)**

   輸入數據點數 NoData，及數據對 X,Y，副程式傳回線性回歸結果 $\hat{y} = \alpha + \beta x$ 之參數 Alpha 及 Beta。

3. 副程式 **Public Sub LinInterpolation(Alpha, Beta)**

   利用回歸分析所得參數，作數據點之內差。

程式執行結果：



函數關係的選擇並非完全沒有規則可依循，對所觀察現象或所量測對象的基本原理之了解，通常有助於建立函數關係。但如本章第一節所討論，實驗數據並不一定都會成線性關係，因此，我們可能時常需面對選擇近似方程式的難題。

若將實驗數據描繪在方格紙上，或用 Excel 軟體嘗試將實驗數據作圖，通常有助於由曲線的形狀及對實驗性質的了解，提出可代表這些數據的方程式。而所提出方程式的適用性，最簡單而有效率的檢查方式，就是先將方程式轉換成線性關係 $\hat{y} = a + b\hat{x}$ ，並作圖判斷數據與方程式的一致性。表 7.6 為一些常見的單變數函數的線性轉換法。利用這種轉換技巧，即可將非線性函數轉換成線性函數，然後再利用本節所介紹的最小平方法推定其係數。

表 7.6　單變數函數的線性轉換

原方程式	直線轉換座標		直線方程式	漸近線及備註
	X 座標	Y 座標		
$\frac{1}{y}=\alpha+\beta x$	x	$1/y$	$Y=\alpha+\beta X$	$X=-\frac{\alpha}{\beta}$　$@Y\to 0$
$y=\alpha+\beta/x$	$1/x$	Y	$Y=\alpha+\beta X$	$Y=\alpha$　$@X\to 0$
$y=\frac{x}{\alpha+\beta x}$	x	x/y	$Y=\alpha+\beta X$	$y=\frac{1}{\beta}$　$@x\to-\frac{\alpha}{\beta}$
	$1/x$	$1/y$	$Y=\beta+\alpha X$	
$y=\frac{x}{\alpha+\beta x}$	x	$\frac{x-x_1}{y-y_1}$	$Y=\frac{a(a+bx_1)}{ad-bc}+\frac{b(a+bx_1)}{ad-bc}X$	$y=\frac{d}{b}$　$@x\to-\frac{a}{b}$
	(x_1,y_1) 爲一數據點			
$y=\alpha x^{\beta}$	$log(x)$	$log(y)$	$Y=\log(\alpha)+\beta X$	$\beta>0$爲拋物線 $\beta<0$爲雙曲線
$y=\alpha x^{\beta}+\gamma$	$log(x)$	$log(y-\gamma)$	$Y=\log(\alpha)+\beta X$	γ之近似值 可利用 $x\to 0$ 求之
$y=\alpha\beta^{x}$	x	$log(y)$	$Y=\log(\alpha)+$ $+\log(\beta)X$	通過　$\alpha(0,\alpha)$

第三節　非線性函數近似法

Visual Basic

上一節中，我們介紹許多種函數線性化的策略，但除了表 7.6 所示的非線性方程式以外，多項式方程式也是一種常被使用的近似方程式。尤其是要對所獲得的近似方程式作微分或積分運算時，多項式更是易於使用。

由於多項式運算容易，因此，作圖結果如果不是直線的函數，即可考慮利用多項式方程式來表示。事實上，我們常用的物理化學性質資料如熱容量 C_p 及黏度 μ 等，通常就是利用多項式做爲近似表示式。

利用多項式函數做爲數據點之最佳近似，其處理策略與上一節所介紹的最小平方

法相同。假設多項式近似函數為

$$\hat{y} = a_0 + a_1 x + a_2 x^2 + \cdots\cdots + a_n x^n = \sum_{i=0}^{n} a_i x^i \tag{7-3.1}$$

定義實驗數據與近似式的誤差為

$$\begin{aligned} e_i &= y_i - \hat{y}_i \\ &= y_i - (a_0 + a_1 x_i + a_2 x_i^2 + \cdots\cdots + a_n x_i^n) \\ &= y_i - \sum_{j=0}^{n} a_j x_i^j \end{aligned} \tag{7-3.2}$$

上式中，y_i 為實驗值，$\hat{y}_i$ 表示近似值；x_i 為自變數，且假設沒有測量誤差。我們希望使誤差的平方和達到最小，

$$\begin{aligned} S &= \sum_{i=1}^{N} e_i^2 \\ &= \sum_{i=1}^{N} [y_i - \sum_{j=0}^{n} a_j x_i^j]^2 \\ &= \sum_{i=1}^{N} [y_i - a_0 - a_1 x_i - a_2 x_i^2 - \cdots\cdots - a_n x_i^n]^2 \end{aligned} \tag{7-3.3}$$

其中 n 為多項式的階次，N 為數據點的總數目。

當 S 值達到最小的時候，所有參數 a_i 對 S 的偏微分應該都等於零。亦即 $\partial S / \partial a_0 = \partial S / \partial a_1 = \partial S / \partial a_2 = \cdots\cdots = \partial S / \partial a_n = 0$，可建立 $n+1$ 個方程式：

$$\frac{\partial S}{\partial a_0} = 0 = \sum_{i=1}^{N} (-2)[y_i - \sum_{j=0}^{n} a_j x_i^j]$$

$$\frac{\partial S}{\partial a_1} = 0 = \sum_{i=1}^{N} (-2x_i)[y_i - \sum_{j=0}^{n} a_j x_i^j]$$

$$\vdots$$

$$\frac{\partial S}{\partial a_n} = 0 = \sum_{i=1}^{N} (-2x_i^n)[y_i - \sum_{j=0}^{n} a_j x_i^j]$$

將各方程式除以 (-2)，展開得到 $n+1$ 個聯立方程式：

$$a_o N + a_1 \sum x_i + a_2 \sum x_i^2 + \cdots\cdots + a_n \sum x_i^n = \sum y_i$$

$$a_o \sum x_i + a_1 \sum x_i^2 + a_2 \sum x_i^3 + \cdots\cdots + a_n \sum x_i^{n+1} = \sum x_i y_i$$

$$a_o \sum x_i^2 + a_1 \sum x_i^3 + a_2 \sum x_i^4 + \cdots\cdots + a_n \sum x_i^{n+2} = \sum x_i^2 y_i$$

$$\vdots$$

$$a_o \sum x_i^n + a_1 \sum x_i^{n+1} + a_2 \sum x_i^{n+2} + \cdots\cdots + a_n \sum x_i^{2n} = \sum x_i^n y_i \qquad \textbf{(7-3.4)}$$

將上列聯立方程式寫成矩陣型式，得

$$\underline{\underline{X}}\ \underline{\underline{X}}^T \underline{A} = \underline{\underline{X}}\ \underline{Y} \qquad \textbf{(7-3.5)}$$

其中

$$\underline{Y}^T = [y_1 , y_2 , \cdots\cdots , y_N]$$

$$\underline{A}^T = [a_0 , a_1 , \cdots\cdots , a_n] \qquad \textbf{(7-3.6)}$$

$$\underline{\underline{X}} = \begin{bmatrix} x_1^0 & x_2^0 & \cdots & x_N^0 \\ x_1^1 & x_2^2 & \cdots & x_N^1 \\ \vdots & \vdots & \cdots & \vdots \\ x_1^n & x_2^n & \cdots & x_N^n \end{bmatrix}$$

方程式 (7-3.4) 可以利用本書第四章所介紹的高斯消去法求解向量 $\underline{A}$，再代回方程式 (7-3.1)，即可得到所需要的近似方程式。

例題 7-3　水的熱容量

表 7.7 是水的熱容量與溫度的關係，通常 C_p 與溫度關係都是利用三次多項式表示。

$$C_P = A + BT + CT^2 + DT^3$$

試利用最小平方法求出係數 A , B , C 及 D。

解：

由圖 7.11 可以看出水的熱容量與溫度的函數關係為非線性關係。但其變化呈現一種平滑的關係，適合利用多項式作為近似方程式。

表 7.7　水的熱容量

#	溫度 T(°C)	熱容量 C_p
1	-	1.00762
2	5.0	1.00392
3	10.0	1.00153
4	15.0	1.00000
5	20.0	0.99907
6	25.0	0.99852
7	30.0	0.99826
8	35.0	0.99818
9	40.0	0.99828
10	45.0	0.99849
11	50.0	0.99878
12	55.0	0.99919
13	60.0	0.99967
14	65.0	1.00024
15	70.0	1.00091
16	75.0	1.00167
17	80.0	1.00253
18	85.0	1.00351
19	90.0	1.00461
20	95.0	1.00586
21	100.0	1.00721



圖 7.11　水的比熱與溫度之函數關係

符號說明：

A：　$\sum_{i=1}^{N} X_i^{\,j-1}$

D：　多項式階次

N：　數據點的數目

X，Y：數據

R：　矩陣　$[\underline{X}\,\underline{X}^T / \underline{X}\,\underline{Y}]$

程式列印：

```
'****************************************
'    POLYNOMIAL CURVE FITTING
'****************************************
'
Private Sub PolynomialRegression(Xpos, Ypos)
Dim A(101), R(51, 52), T(52)
Dim X(100), Y(100), Xhold(100), Yhold(100)
'
' INPUT DATA:
' N=NO. OF DATA POINTS
' D=ORDER OF THE MODEL
' X(I),Y(I)=DATA POINT
'
' A DEMO PROGRAM
Dmax = 50
N = 0
Ent% = 0
SelID% = 1
Do
    Call Menu(N, Ent%, SelID%, Xpos, Ypos)
    Select Case SelID%
    Case 1
        Cls
        Print "ENTER DATA POINTS:"
        Call DataInput(N, X, Y, Xhold, Yhold, Ent%, Xpos, Ypos)
        A(1) = N
        Call DataTransform(N, X, Y, Xhold, Yhold)
    Case 2
        Call GetModelOrder(D, N, Dmax, Xpos, Ypos)
        Flag% = 1
        Call PolyRegression(D, X, Y, N, A, T, R, Flag%)
        If Flag% = 0 Then
            Print "ERR-"
            MsgBox ("Data Error, Regression Terminated")
        Else
            Call Report(D, R, N, X, Y, A, T)
            Call InterCal(D, R, Xpos, Ypos)
        End If
    Case 3
        Exit Do
    End Select
Loop While Ent% = 1 Or Ent% = 2
End Sub
```

```
Private Sub DataTransform(N, X, Y, Xhold, Yhold)
'
' for Data Transform or Linearization
'
' 1. Keep orginal data for further processing
'
For I = 1 To N
        Xhold(I) = X(I)
        Yhold(I) = Y(I)
Next I
'
' 2. Data transform, to be defined by User
'


End Sub


Public Sub PolyRegression(D, X, Y, N, A, T, R, Flag%)
'
' POLYNOMIAL CURVE FITTING
'
' INPUT DATA:
' D=ORDER OF THE MODEL
' X(I),Y(I)=DATA POINT
' N=NO. OF DATA POINTS
'
If D > (N - 1) Then
        Flag% = 0
        Exit Sub
End If

For I = 2 To 2 * D + 1
        A(I) = 0
Next

For I = 1 To D + 2
        T(I) = 0
Next
'
' REG ROUTINE ENTRY POINT
'
Flag% = 0
For I = 1 To N
        For J = 2 To 2 * D + 1
                V = J - 1
                A(J) = A(J) + X(I) ^ V
```

```
        Next J
        For K = 1 To D + 1
            R(K, D + 2) = T(K) + Y(I) * X(I) ^ (K - 1)
            T(K) = T(K) + Y(I) * X(I) ^ (K - 1)
        Next K
        T(D + 2) = T(D + 2) + Y(I) ^ 2
    Next I
    '
    For J = 1 To D + 1
        For K = 1 To D + 1
            R(J, K) = A(J + K - 1)
        Next K
    Next J
    '
    'Gaussian Elimination
    '
    For J = 1 To D + 1
        For K = J To D + 1
            If R(K, J) <> 0 Then
                Flag% = 1
                Exit For
            End If
        Next K

        If Flag% <> 1 Then
            Flag% = 0
            Print "NO UNIQUE SOLUTION"
            Exit Sub
        End If

        For I = 1 To D + 2
            Call Swap(R(J, I), R(K, I))
        Next I

        Z = 1 / R(J, J)

        For I = 1 To D + 2
            R(J, I) = Z * R(J, I)
        Next I

        For K = 1 To D + 1
            If K <> J Then
                Z = -R(K, J)
                For I = 1 To D + 2
                    R(K, I) = R(K, I) + Z * R(J, I)
                Next I
            End If
```

```
        Next K
Next J
End Sub


Public Sub Report(D, R, N, X, Y, A, T)
Cls
Print "*******REGRESSION ANALYSIS *******"
Print "                            N "
Print " MODEL:     Y = SUM C(I) * X ^ I"
Print "                          I=0"
Print "                                  CONSTANT =";
Print Format(R(1, D + 2), " 0.000000E+00")
For J = 1 To D
        Print Format(J, " ###     DEGREE COEEFFICIENT =");
        Print Format(R(J + 1, D + 2), " 0.000000E+00")
Next J
Print "======================================================="
Print "       #                    X                          Y                          Y(CALC)"
Print "-----------------------------------------------------"
For I = 1 To N
        Print Format(I, " 000");
        Print Format(X(I), "        0.000000E+00");
        Print Format(Y(I), "        0.000000E+00");
        P = R(1, D + 2)
        For J = 1 To D
                P = P + R(J + 1, D + 2) * X(I) ^ J
        Next J
        Print Format(P, "        0.000000E+00")
Next I
Print "======================================================="

MsgBox ("Hit Button to Continue")
Cls
P = 0
For J = 2 To D + 1
        P = P + R(J, D + 2) * (T(J) - A(J) * T(1) / N)
Next J
Q = T(D + 2) - T(1) ^ 2 / N
Z = Q - P
I = N - D - 1
J = P / Q
Print "COEFFICIENT OF DETERMINATION (R^2) =";
Print Format(J, " 0.000000E+00")
Print "CORRELATION COEFFICIENT =";
Print Format(Sqr(Abs(J)), " 0.000000E+00")
Print "STANDARD ERROR ESTIMATE =";
```

```
Print Format(Sqr(Abs(Z / I)), " 0.000000E+00")
MsgBox ("Hit Button to Continue")
End Sub


Public Sub InterCal(DN, R, Xpos, Ypos)
'
' ===ESTIMATE Y-COORDINATES OF POINTS WITH ENTERED X-COORDINATES===
'
Cls
Print "INTERPOLATION: "
Print
Do
        Print "X = ";
        X = Val(InputBox("Enter X Value for Interpolation", "X", X, Xpos, Ypos))
        Print Format(X, " 0.000000E+00         ");

        P = R(1, DN + 2)
        For J = 1 To DN
                P = P + R(J + 1, DN + 2) * X ^ J
        Next J

        Print Format(P, "     0.000000E+00")
        NextDataYN$ = InputBox("Next Data <Y/N> ?", "Y/N", "N", Xpos, Ypos)
Loop While NextDataYN$ <> "N"
End Sub


Public Sub Menu(N, Ent%, SelID%, Xpos, Ypos)
'
' Menu for Polynomial Curve Fitting
'
If N > 0 And Ent% = 1 Then
        Cls
        Print
        Print "<1> Run New Job"
        Print "<2> Chang the Order of Polynomial"
        Print "<3> Exit"
        SelID% = Val(InputBox("Select Job ID", "Job ID", 2, Xpos, Ypos))
ElseIf N = 0 Then
        SelID% = 1
ElseIf N > 0 Then
        SelID% = 2
        Ent% = 1
End If
End Sub
```

```
Public Sub DataInput(N, X, Y, Xhold, Yhold, Ent%, Xpos, Ypos)
If N > 0 Then Call GetDataBack(N, X, Y, Xhold, Yhold)
N = 0
Do
        N = N + 1
        Print "Enter X("; N; ") = ";
        X(N) = Val(InputBox("Enter X Value = ", "X", X(N), Xpos, Ypos))
        Print X(N);

        Print "            Y("; N; ") = ";
        Y(N) = Val(InputBox("Enter Y Value = ", "Y", Y(N), Xpos, Ypos))
        Print Y(N)

        NextDataYN$ = InputBox("Next Data <Y/N>", "NEXT DATA", "Y", Xpos, Ypos)
Loop While NextDataYN$ = "Y"
Ent% = 2
End Sub


Public Sub GetModelOrder(D, N, Dmax, Xpos, Ypos)
Do
      D = Val(InputBox("ORDER OF THE MODEL (<=50) = ", "Order of Polynomial", D, Xpos, Ypos))
      If D > N - 1 Then
                Print "Reduce Order D, Please!"
                Print "Maximum Order Dmax = "; N - 1
                MsgBox ("Reduce D")
        End If
Loop While D > Dmax
End Sub


Public Sub GetDataBack(N, X, Y, Xhold, Yhold)
For I = 1 To N
        X(I) = Xhold(I)
        Y(I) = Yhold(I)
Next I
End Sub


Public Sub Swap(Z1, Z2)
Z3 = Z1
Z1 = Z2
Z2 = Z3
End Sub
```

副程式使用說明：

1. 副程式 **Public Sub Menu(N, Ent%, SelID%, Xpos, Ypos)**
   用於取得工作選項 SelID%，Xpos 及 Ypos 爲對話方塊定位位址。

2. 副程式 **Private Sub DataTransform(N, X, Y, Xhold, Yhold)**
   供使用者作數據轉換的空間，將原始數據 (X，Y) 作轉換後，存回 (X，Y) 位址，未轉換的數據將暫存於 (Xhold, Yhold)。

3. 副程式 **Public Sub GetDataBack(N, X, Y, Xhold, Yhold)**
   供使用者將未轉換的數據 (Xhold , Yhold) 取回至 (X , Y)。

4. 副程式 **Public Sub GetModelOrder(D, N, Dmax, Xpos, Ypos)**
   供使用者取得多項式方程式的階次 D。

5. 副程式 **Public Sub PolyRegression(D, X, Y, N, A, T, R, Flag%)**
   將 N 組數據點 (X，Y) 作 D 次多項式迴歸分析，並將所得結果存放在 R，其錯誤訊息放在 Flag%。Flag%=0 表示計算錯誤或輸入錯誤。

6. 副程式 **Public Sub Report(D, R, N, X, Y, A, T)**
   列印分析結果，並計算(X,Y)與計算值之比較。

7. 副程式 **Public Sub InterCal(DN, R)**
   供使用者輸入 X 值，計算 Y 值。

程式執行結果：

將表 7.6 的數據逐筆輸入，輸入完成後，在對話方塊詢問是否繼續輸入數據時，回答 N 就可結束數據輸入。事實上，電腦最強的功能就是計算，因此，數據究竟有多少筆，理應由電腦自行計算，本程式就是一個典型的程式設計方式。另外，本程式也具有輸入數據更正的功能，及變更近似方程式的功能。

多項式階次選擇使用 3，所得結果如下：

```
Polynomial Regression
*******REGRESSION ANALYSIS *******
              N
 MODEL:  Y = SUM C(I) * X ^ I
             I=0
                 CONSTANT = 1.006446E+00
 1  DEGREE COEEFFICIENT =- 4.985711E-04
 2  DEGREE COEEFFICIENT = 8.453810E-06
 3  DEGREE COEEFFICIENT =- 3.453390E-08
======================================================
   #          X               Y               Y(CALC)
------------------------------------------------------
 001    0.000000E+00    1.007620E+00    1.006446E+00
 002    5.000000E+00    1.003920E+00    1.004160E+00
 003    1.000000E+01    1.001530E+00    1.002271E+00
 004    1.500000E+01    1.000000E+00    1.000753E+00
 005    2.000000E+01    9.990700E-01    9.995794E-01
 006    2.500000E+01    9.985200E-01    9.987253E-01
 007    3.000000E+01    9.982600E-01    9.981644E-01
 008    3.500000E+01    9.981800E-01    9.978709E-01
 009    4.000000E+01    9.982800E-01    9.978186E-01
 010    4.500000E+01    9.984900E-01    9.979819E-01
 011    5.000000E+01    9.987800E-01    9.983348E-01
 012    5.500000E+01    9.991900E-01    9.988514E-01
 013    6.000000E+01    9.996700E-01    9.995057E-01
 014    6.500000E+01    1.000240E+00    1.000272E+00
 015    7.000000E+01    1.000910E+00    1.001124E+00
 016    7.500000E+01    1.001670E+00    1.002036E+00
 017    8.000000E+01    1.002530E+00    1.002983E+00
 018    8.500000E+01    1.003510E+00    1.003938E+00
 019    9.000000E+01    1.004610E+00    1.004875E+00
 020    9.500000E+01    1.005860E+00    1.005768E+00
 021    1.000000E+02    1.007210E+00    1.006593E+00
======================================================
```

Polynomial Regression

Hit Button to Continue

確定

CHEER   Copyright RESI 2001

開始執行

```
Polynomial Regression
COEFFICIENT OF DETERMINATION (R^2) = 9.740892E-01
CORRELATION COEFFICIENT = 9.869596E-01
STANDARD ERROR ESTIMATE = 5.312440E-04
```

Polynomial Regression

Hit Button to Continue

確定

CHEER   Copyright RESI 2001

開始執行

亦即水的熱容量與溫度關係可寫成

$$C_p = 1.006446 - 0.00049857T + 0.0000084538T^2 - 0.0000000345\,34T^3$$

第四節　多變數線性迴歸

Visual Basic

實驗數據時常會含有兩個以上的變數。例如氫氣與氮氣經催化生成氨的反應常數 K_p，即決定於反應溫度 T 及壓力 P，亦即 $K_p = f(T, P)$。其次，如流體在管中流動的對流熱傳係數與流體流速、物性及管徑有關，利用因次分析可得到以下的關係式：

$$Nu = f(\mathrm{Re}, \mathrm{Pr})$$

其中 $Nu = hD/k$, $\mathrm{Re} = DU\rho/\mu$, $\mathrm{Pr} = \rho C_p \mu/k$, h 爲熱傳係數，D 爲管子的直徑，k 爲流體的熱傳導係數，ρ 爲流體密度，μ 爲流體的黏度，C_p 爲流體的熱容量，U 爲流體的平均流速。

因此，在分析實驗數據時，時常需將一變數寫成其他變數的函數，如

$$v = f(x_1, x_2, \cdots\cdots, x_m) \tag{7-4.1}$$

在本節中，我們將只線討論線性模式的迴歸分析，其他型式如多項式、指數函數、三角函數等亦可利用相似的方法處理之。假設函數 f 爲 x_i，$i = 1, 2, \ldots\ldots, m$ 的線性函數，則方程式 (7-4.1) 可改寫成

$$y = a_0 + a_1 x_1 + a_2 x_2 + \ldots\ldots + a_m x_m = \sum_{i=0}^{m} a_i x_i \quad ; \quad x_0 = 1 \tag{7-4.2}$$

其中 $\hat{y}$ 爲模式推定值。線性模式推定值 $\hat{y}$ 與實驗測量值 y 的誤差爲

$$e_i = y_j - \hat{y}_j = y_j - \sum_{i=0}^{m} a_i x_{ij} \quad ; \quad j = 1, 2, \cdots\cdots, N \tag{7-4.3}$$

其中 x_{ij} 表示第 i 個變數 x_i 的第 j 個測量數值。由上式得誤差的平方和爲

$$S = \sum_{j=1}^{N} \left(y_j - \sum_{i=0}^{m} a_i x_{ij}\right)^2 \tag{7-4.4}$$

若以矩陣符號 $\underline{Y}$, $\underline{A}$ 及 $\underline{X}$ 分別表示 y_i, a_i 及 x_{ij}，

$$\underline{Y}^T = [y_1, y_2, y_3, ..., y_N]$$

$$\underline{A}^T = [a_1, a_2, a_3, ..., a_m]$$

$$\underline{\underline{X}} = \begin{bmatrix} x_{11} & x_{12} & x_{13} ...... x_{1N} \\ x_{21} & x_{22} & x_{23} ...... x_{2N} \\ \vdots & & \vdots \\ \vdots & & \vdots \\ x_{N1} & x_{N2} & x_{N3} ...... x_{NN} \end{bmatrix}$$

則誤差的平方和可寫成矩陣符號

$$S = (\underline{Y} - \underline{\underline{X}}^T \underline{A})^2 \tag{7-4.5}$$

S 達最小值的條件為 $\partial S / \partial \underline{A} = 0$ ，即

$$\frac{\partial S}{\partial \underline{A}} = -2\underline{\underline{X}}(\underline{Y} - \underline{\underline{X}}^T \underline{A}) = -2(\underline{\underline{X}}\,\underline{Y} - \underline{\underline{X}}\,\underline{\underline{X}}^T \underline{A}) = 0 \tag{7-4.6}$$

故可得矩陣方程式

$$\underline{\underline{X}}\,\underline{\underline{X}}^T \underline{\underline{A}} = \underline{\underline{X}}\,\underline{Y} \tag{7-4.7}$$

求解待定係數向量或近似模式之參數值得

$$\underline{A} = (\underline{\underline{X}}\,\underline{\underline{X}}^T)^{-1} \underline{\underline{X}}\,\underline{Y} \tag{7-4.8}$$

將方程式 (7-4.8) 所得的係數 $\underline{A}^T = [a_1, a_2, \cdots\cdots, a_m]$ 代回方程式 (7-4.2)，即得 y 對 x_1 的線性迴歸方程式。

例題 7-4 管內的強制對流傳熱（設計問題 VII)

強制對流熱傳問題的實驗數據分析，通常寫成

$$Nu = C_1 \,\mathrm{Re}^{\alpha}\, \mathrm{Pr}^{\beta} \left(\frac{L}{D}\right)^{\gamma} \left(\frac{\mu_b}{\mu_s}\right)^{\delta}$$

將此方程式左右兩側分別取對數，即可成為一個多變數的線性模式

$$\mathrm{Y} = \mathrm{A}_0 + \mathrm{A}_1 \mathrm{X}_1 + \mathrm{A}_2 \mathrm{X}_2 + \mathrm{A}_3 \mathrm{X}_3 + \mathrm{A}_4 \mathrm{X}_4$$

其中

$$Y = \ln(Nu)\,;\, X_1 = \ln(\mathrm{Re})\,;\, X_2 = \ln(\mathrm{Pr})\,;\, X_3 = \ln\left(\frac{L}{D}\right)$$

$$X_4 = \ln\left(\frac{\mu_b}{\mu_5}\right);\, A_0 = \ln(C_1)\,;\, A_1 = \alpha\,,\, A_2 = \beta\,,\, A_3 = \gamma\,,\, A_4 = \delta$$

設計程式時，先將實驗數據取對數後，再作線性迴歸分析。

由於多變數線性模式方程式 (7-4.7) 及多項式模式方程式 (7-3.5) 可得到完全相同型式的聯立方程式，因此，程式設計時可考慮將二者合而為一，考慮一般性的多變數多項式模式迴歸分析。

符號說明：

A (I , J)：係數矩陣 $\underline{\underline{X}}\,\underline{\underline{X}}^T$

B (I)：　矩陣 $\underline{\underline{X}}\,\underline{Y}$

D (I , J)：操作矩陣[$\underline{\underline{X}}\,\underline{\underline{X}}^T$ | —]，用於高斯消去法

IP：　參數的個數

M：　模式的最高次數

N：　自變數的個數

NN：　數據點數

PAR：　參數值

X：　x_{ij}

Y：　Y_j

YEST：　Y_j 的模式推測值 $\hat{y}_j$

程式列印：

```
'******************************************
' Multivariable/Polynomila Regression
'******************************************
'
Private Sub MultiPolynomialRegression(Xpos, Ypos)
Dim X(50, 10), Y(50)
Dim Xhold(50, 10), Yhold(50)
Dim Dmatrix(21, 21), A(21, 21), B(21)
Dim Parameter(21), Yest(50)
'
' N         = NO. IND. VAR.
' NN        = NO. DATA SET
```

```
' M         = Order of model.     Max. = 3
' X         = Independent Variable
' Y         = Dependent Variable
'
Cls
Print "Number of Independent Variables = ";
N = Val(InputBox("Number of independent variables = ", "N", N, Xpos, Ypos))
NN = 0
Ent% = 0
SelID% = 1
Do
        Call Menu(NN, Ent%, SelID%, Xpos, Ypos)
        Select Case SelID%
        Case 1
                Call DataInput(N, NN, X, Y, Xhold, Yhold, Xpos, Ypos)
                Ent% = 2
'
'==         Linearization of data by USER
'
                Call DataTransform(N, NN, X, Y, Xhold, Yhold)
        Case 2
                M = Val(InputBox("ORDER OF THE MODEL = ", "Order of Polynomial", M, Xpos, Ypos))
                Flag = 1
                Call MultiPolyRegression(N, NN, M, X, Y, A, B, Dmatrix, Parameter, Yest, Flag)
                If Flag = 0 Then
                        Print "ERR-"
                        MsgBox ("Data Error, Regression Terminated")
                End If
        Case 3
                Exit Do
        End Select
Loop While Ent% = 1 Or Ent% = 2
End Sub


Private Sub DataLinearize(N, NN, X, Y)
'
' N         = NO. IND. VAR.
' NN        = NO. DATA SET
'
' X         = Independent Variable
'                   Input raw data
'                   Returned linearized data
' Y         = Dependent Variable
'                   Input raw data
'                   Returned linearized data
'
```

```
' ****************** USER'S MODEL HERE
For I = 1 To NN
        For K = 2 To N + 1
                X(I, K) = Log(X(I, K))
        Next K
        Y(I) = Log(Y(I))
Next I
End Sub


Private Sub DataTransform(N, NN, X, Y, Xhold, Yhold)
'
' for Data Transform or Linearization
'
' 1. Keep orginal data for further processing
'
For I = 1 To NN
        For J = 1 To N + 1
                Xhold(I, J) = X(I, J)
        Next J
        Yhold(I) = Y(I)
Next I
'
' 2. Data transform
'
Call DataLinearize(N, NN, X, Y)
End Sub


Public Sub GetDataBack(N, NN, X, Y, Xhold, Yhold)
For I = 1 To NN
        For J = 1 To N + 1
                X(I, J) = Xhold(I, J)
        Next J
        Y(I) = Yhold(I)
Next I
End Sub


Public Sub Menu(NN, Ent%, SelID%, Xpos, Ypos)
If NN > 0 And Ent% = 1 Then
        Cls
        Print
        Print "<1> Run New Job"
        Print "<2> Chang/Enter the Order of Polynomial"
        Print "<3> Exit"
        SelID% = Val(InputBox("Select Job ID", "Job ID", 2, Xpos, Ypos))
```

```
ElseIf NN = 0 Then
        SelID% = 1
ElseIf NN > 0 Then
        SelID% = 2
        Ent% = 1
End If
End Sub


Public Sub DataInput(N, NN, X, Y, Xhold, Yhold, Xpos, Ypos)
If NN > 0 Then Call GetDataBack(N, NN, X, Y, Xhold, Yhold)
NN = 0
Cls
Print "ENTER DATA POINTS:"
Do
        NN = NN + 1
        X(NN, 1) = 1!
        For J = 2 To N + 1
                Print "X("; J - 1; ","; NN; ") = ";
                X(NN, J) = Val(InputBox("Enter X Value = ", "X", X(NN, J), Xpos, Ypos))
                Print X(NN, J)
        Next J
        Print "            Y("; NN; ") = ";
        Y(NN) = Val(InputBox("Enter Y Value = ", "Y", Y(NN), Xpos, Ypos))
        Print Y(NN)
        NextDataYN$ = InputBox("Next Data <Y/N>", "NEXT DATA", "Y", Xpos, Ypos)
        If (Int(NN / 3) * 3 = NN) Then Cls
Loop While NextDataYN$ = "Y"
End Sub


Public Sub MultiPolyRegression(N, NN, M, X, Y, A, B, Dmatrix, Parameter, Yest, Flag)
'
' N         = NO. IND. VAR.
' NN        = NO. DATA SET
' M         = Order of model.    Max. = 3
' X         = Independent Variable
' Y         = Dependent Variable
'
' --CHECK Multi/Poly
'
If N < 1 Then
        IP = 1
ElseIf N > 1 Then
        If M > 3 Then
                Cls
                Print "DEU TO THE LIMITATION OF MEMORY "
```

```
                Print "CAPACITY, MAXIMUM ALLOWABLE ORDER IS 3"
                Print "PLEASE USE ANOTHER MODEL"
                Print
                Call CheckOut(Decision$)
                MsgBox ("Ready to Check out from The Program")
                If Decision$ = "C" Then Exit Sub
        Else
'
' == START ==
'
                IP = N + 1
                If M > 1 Then
                                Call PackSecondOrder(N, NN, IP, X)
                        If M > 2 Then
                                Call PackThirdOrder(N, NN, IP, X)
                        End If
                End If
        End If
Else
        If M <= 1 Then
                IP = M + 1
        Else
                For I = 1 To NN
                        For J = 2 To M
                                X(I, J + 1) = X(I, 2) ^ J
                        Next J
                  Next I
                  IP = M + 1
        End If
End If

ZZ = 2 * IP + 2
'
For I = 1 To IP
        B(I) = 0
        For J = 1 To NN
                B(I) = B(I) + X(J, I) * Y(J)
        Next J
        For J = 1 To IP
                A(I, J) = 0
                For K = 1 To NN
                        A(I, J) = A(I, J) + X(K, I) * X(K, J)
                Next K
        Next J
Next I
'
Z1 = IP + 2
```

```
For I = 1 To IP
        For J = 1 To ZZ
                Dmatrix(I, J) = 0
        Next J
        For J = 1 To IP
                Dmatrix(I, J) = A(I, J)
        Next J
        Dmatrix(I, IP + 1) = 0
Next I
For J = Z1 To ZZ
        Dmatrix(J - IP - 1, J) = 1
Next J
DET = 1
For K = 1 To IP
        DET = DET * Dmatrix(K, K)
        If (Abs(Dmatrix(K, K)) < 0.0000000001) Then
                Print K, Dmatrix(K, K)
                Print "SMALL PIVOT ........MATRIX MAY BE SINGULAR"
                Flag = 0
                Call CheckOut(Decision$)
                MsgBox ("Ready to Check out from The Program")
                If Decision$ = "C" Then Exit Sub
        End If
        For J = K + 1 To ZZ
                Dmatrix(K, J) = Dmatrix(K, J) / Dmatrix(K, K)
        Next J
        Dmatrix(K, K) = 1
        For I = 1 To IP
                If I <> K And Dmatrix(I, K) <> 0 Then
                        For J = K + 1 To ZZ
                                Dmatrix(I, J) = Dmatrix(I, J) - Dmatrix(I, K) * Dmatrix(K, J)
                        Next J
                        Dmatrix(I, K) = 0
                End If
        Next I
Next K
For I = 1 To IP
        For J = Z1 To ZZ
                A(I, J - IP - 1) = Dmatrix(I, J)
        Next J
Next I
'
' == PARAMETER ==
'
For I = 1 To IP
        Parameter(I) = 0
        For J = 1 To IP
```

```
                Parameter(I) = Parameter(I) + A(I, J) * B(J)
        Next J
Next I
'
' == STATISTICS ==
'
For I = 1 To NN
        Yest(I) = 0
        For K = 1 To IP
                Yest(I) = Yest(I) + X(I, K) * Parameter(K)
        Next K
Next I
'
' == OUTPUT ==
'
Sum = 0
For I = 1 To NN
        Sum = Sum + (Y(I) - Yest(I)) ^ 2
Next I

Var = Sum / (NN - IP)
StandardDeviation = Sqr(Var)
Cls
Print
Print "****************************************************"
Print "NUMBER OF INDEPENDENT VARIABLES = "; N
Print "ORDER OF THE MODEL = "; M
Print "NUMBER OF PARAMENTERS = "; IP
Print
Print "***** X(J,I)............Y(I) & Ycal(I)"
For I = 1 To NN
        For J = 2 To N + 1
                Print Format(X(I, J), " 0.0000E+00     ");
        Next J
        Print Format(Y(I), "     0.0000E+00");
        Print Format(Yest(I), "       0.0000E+00")
Next I
MsgBox ("Ready to Continue")
Cls
Print
Print ".........STATISTICS........."
Print "RESIDUAL MEAN SQUARE = "; Var
Print "SQRT(VAR) = "; StandardDeviation
Print
Print "PARAMETERS:"
Print
For I = 1 To IP
```

```
        Print "Parameter("; (I - 1); ") = ";
        Print Format(Parameter(I), " 0.000000E+00     ");
        Print "+/-"
        Print Format(A(I, I) * StandardDeviation, "     0.000000E+00")
Next I
Print
Print "***********************************************************"
Print "END OF REGRESSION ANALYSIS "
Print
Print "C.H.E.E.R. - 2001 By Dr. Ron Hsin Chang"
MsgBox ("Ready to Continue")
End Sub


Public Sub PackSecondOrder(N, NN, IP, X)
For I = 2 To N + 1
        For J = I To N + 1
                IP = IP + 1
                For K = 1 To NN
                        X(K, IP) = X(K, I) * X(K, J)
                Next K
        Next J
Next I
End Sub


Public Sub PackThirdOrder(N, NN, IP, X)
For I = 1 To N + 1
        For J = I To N + 1
                For K = J To N + 1
                        IP = IP + 1
                        For L = 1 To NN
                                X(L, IP) = X(L, I) * X(L, J) * X(L, K)
                        Next L
                Next K
        Next J
Next I
End Sub


Public Sub CheckOut(Decision$)
        Print "A)BORT OR (C)ONTINUE";
        Decision$ = InputBox("A)bort or C)ontinue ", "     ", , Xpos, Ypos)
        If Decision$ = "A" Then End
        If Decision$ = "C" Then Exit Sub
End Sub
```

程式執行結果：

將設計問題 VII 的數據輸入後，所得結果如下圖所示。由於原方程式參數為 $A_0 = \ln(C_1)$；$A_1 = \alpha$；$A_2 = \beta$；$A_3 = \gamma$，$A_4 = \delta$；因此，需將結果處理，以得到正確常數值，$C_1 = Exp(0.7495208) = 2.116$。經最小平方法迴歸分析，所得方程式為

$$Nu = 2.116\,\mathrm{Re}^{0.329}\,\mathrm{Pr}^{0.333}\left(\frac{L}{D}\right)^{-0.356}\left(\frac{\mu_b}{\mu_s}\right)^{0.156}$$



實驗所得那塞數 Nu 與上式右側計算值的比較，如圖 7.12 所示，其橫座標 $F = 2.116\,\mathrm{Re}^{0.320}\,\mathrm{Pr}^{0.333}\left(\frac{L}{D}\right)^{-0.358}\left(\frac{\mu_b}{\mu_s}\right)^{0.156}$。圖 7-12 的作圖方式，是研究工作者時常用於比較理論分析結果與實驗測量值間的誤差的典型做法，若二者相吻合則所有數據點應位於 Y = X 的對角線上。如圖示，此分析結果相當良好，與文獻上常見的方程式極為接近。

$$Nu = 1.86\left[\mathrm{Re}\cdot\mathrm{Pr}\cdot\left(\frac{D}{L}\right)\right]^{\frac{1}{3}}\left(\frac{\mu_b}{\mu_s}\right)^{0.14}$$

當 F 值極低時，*Nu* 會趨近於 3.66。圖中左下角數點即有此傾向，經初步分析發覺此現象，可先將最左下方的三點去除，再進行迴歸分析，可得較佳結果。



圖 **7.12**　經驗方程式與實驗所得 Nu 之比較

使用本程式時，若所使用模式爲多變數線性模式

$$y = a_0 + a_1 x_1 + a_2 x_2 + \cdots\cdots + a_n x_n$$

則需將本程式中的副程式 *DataLinearize(N, NN, X, Y)*中之變數轉換取消，即可直接使用。本程式亦適用於多項式模式及多變數齊次多項式模式

$$y = a_0 + a_1 x + a_2 x^2 + \cdots\cdots + a_m x^m$$

及多變數多項式模式

$$y = a_0 + a_1 x_1 + a_2 x_2 + a_3 x_1^2 + a\ \ x_1 x_2 + a_5 x_2^2 + \cdots\cdots$$

參考文獻

1. Whitaker, S., “Elementary Heat Transfer Analysis,” (1976).

2. Himmelblau, D.M., “Process Analysis by Statistical Methods” (1969).

3. Gerald, C.F., “Applied Numerocal Analysis” 3rd ed., Addison-Wesley, (1984)

4. 劉清田、王逢盛，”計算機套用程式—化學工程教學用程式”台北，(1982).

習　題

1. 利用三角函數模式

$$y = a_0 + a_1 \cos x + a_2 \sin x + a_3 \cos 2x + a_4 \sin 2x$$

分析以下的數據：

x	y	x	y
0	3.0	3.0	–1.8
0.3	4.8	3.3	0.
0.6	6.1	3.6	2.0
0.9	6.4	3.9	3.5
1.2	5.5	4.2	4.1
1.5	3.5	4.5	3.7
1.8	1.1	4.8	2.6
2.1	–1.1	5.1	1.4
2.4	–2.5	5.4	0.5
2.7	–3.8	5.7	0.5
3.0	–1.8	6.0	1.5

2. 試利用多項式模式分析以下數據，多項式最高階次分別試使用 1 至 7，並比較及建議判斷最佳近似式的方法。

x_i	0.05	0.11	0.15	0.31	0.46	0.52
y_i	0.956	0.890	0.832	0.717	0.571	0.539
x_i	0.70	0.74	0.82	0.98	1.17	
y_i	0.378	0.370	03.06	0.242	0.104	

3. 下表為高壓狀況下正丁烷在無水氫氟酸中的溶解度數據，此數據在石油重解爐的設計時需使用。試利用各種作圖方法找出一種最可行的數據分析模式，並設計程式加以分析。

溫度 T(°F)	77	100	185	239	285
溶解度（Wt%)	2.4	3.4	7.0	11.1	19.6

4. 某工程師利用水進行管內強制對流的熱傳遞研究，他利用因次分析證實

$$Nu = \alpha \mathrm{Re}^{\beta}$$

其中 Nu 為納塞數，Re 為雷諾數，α 及 β 均為常數。假設 Re 無測量誤差，試根據以下數據找出 α 及 β 的最佳估計值。

Re	100	100	200	200	300	300	400	400	500	500
Nu	32	36	39	41	40	42	43	45	46	49

5. 在一批式完全混合反應器中，進行 N_2O_5 的第一次均勻相反應，則 N_2O_5 的壓力與時間關係可寫成

$$Ln(\frac{P}{P_0}) = k(t - t_0)$$

其中 P 為 N_2O_5 的分壓，P_0 為時間 $t = t_0$ 時的分壓，t_0 為參考時間，k 為待定的反應常數。試根據以下數據估計 k 值。(根據文獻資料 k 約為 0.0078 至 $0.0096\ \mathrm{min}^{-1}$)。

時間 (min)	$P_{N_2O_5}$ (mmHg)
0	308.2
20	254.4
30	235.5
40	218.2
50	202.2
60	186.8
100	137.2
140	101.4
200	63.6

6. 某一材料的結晶溫度 - 時間關係可用下式表示：

$$\frac{1}{t} = Ae^{-E/RT}$$

其中 t 為時間，E 為活化能，R 為氣體常數，T 為溫度。試根據下列數據推測其活化能 E。

計算時請記得將溫度單位轉換成 K，氣體常數請使用 $R = 1.987$ (Cal/°C · mole) 。

【Ans】E = 15.2 ; A = 4470

溫度 T(°C)	時間 t(min)
350	49
360	40
370	34
380	28
390	24
400	20
410	17
420	14
430	12

7. 石墨的熱容量C_p與溫度的關係為

$$C_P = A + BT + C/T^2$$

試利用以下數據推定常數 A、B 及 C。

T(K)	C_p (Cal/K-Mole)
300	2.08
400	2.85
500	3.50
600	4.03
700	4.43
800	4.75
900	4.98
1000	5.14
1100	5.27
1200	5.42

8. 試將以下非線性化學狀態方程式線性化，並試由蒸氣物性表中任取 30 組數據，分別估算各狀態方程式的常數值：

(a) 凡得瓦狀態方程式 Van der Waals Equation:

$$(P + \frac{a}{V^2})(V - b) = RT$$

(b) 克勞希斯狀態方程式 Clausius Equation:

$$P = \frac{RT}{(V-b)} - \frac{a}{T(V+c)^2}$$

(c) 羅倫茲狀態方程式 Lorentz Equation:

$$P = \frac{RT}{V^2}(V+b) - \frac{a}{V^2}$$

(d) 戴特瑞希狀態方程式 Dieterici Equation:

$$P = \frac{RT}{(V-b)} e^{-a/VRT}$$

(e) 伯斯羅特狀態方程式 Berthelot Equation:

$$P = \frac{RT}{(V-b)} - \frac{a}{TV^2}$$

(f) 赫氏狀態方程式 Wohl Equation:

$$P = \frac{RT}{(V-b)} - \frac{a}{V(V-b)} + \frac{c}{TV^3}$$

(g) 貝遜 - 卜瑞其門狀態方程式 Beattie-Bridgeman Equation:

$$PV = RT + \frac{\beta}{V} + \frac{\gamma}{V^2} + \frac{\delta}{V^3}$$

$$\beta = R\,T\ B_0 - A_0 - \frac{R\,c}{T^2}$$

$$\gamma = -RTB_0 b + aA_0 + \frac{RB_0 c}{T^2}$$

$$\delta = \frac{RB_0 bc}{T^2}$$

(h) 卞尼迪克 - 韋博 - 魯濱狀態方程式 Benedict-Webb-Rubin Equation:

$$PV = RT + \frac{\beta}{V} + \frac{\sigma}{V^2} + \frac{\delta}{V^4} + \frac{\omega}{V^5}$$

$$\beta = R\,T\ B_0 - A_0 - \frac{C_0}{T^2}$$

$$\sigma = RTb - a + \frac{c}{T^2} Exp(-\frac{\gamma}{V^2})$$

$$\delta = c\,\gamma\ Exp(-\frac{\gamma}{V^2})$$

$$\omega = a\alpha$$

以上各狀態方程式中所使用的符號為

p =壓力

V=莫耳體積

T=絕對溫度

R=氣體常數

其他符號都是待決定的係數。

9. 下表為二氧化氮被一反應溶液吸收，用於製造反應產物所獲得的數據。試找出一最佳的多項式近似式。

Y (g/Liter) 二氧化氮被吸收量	X (g/Liter) 產物濃度
0.09	15.1
0.32	57.3
0.69	103.3
1.51	174.6
2.29	191.5
3.06	193.2
3.39	178.7
3.63	172.3
3.77	167.5

10. 試利用以下模式重新處理問題 9 的數據，並與問題 9 所得結果作比較。

$$y = a_1 e^{a_2 x} X^{a_3}$$

11. 試利用泰勒級數展開將下列模式線性化，並仿第四節，建立決定常數的聯立方程式。

$$y = a_1 e^{-a_2 x} + a_3 e^{a_4 x}$$

12. 一完全攪拌的反應器，用於實驗取得建設化工廠所需的規模放大數據。此液相反應的化學計量平衡式為

$$A + B \rightarrow P$$

用於描述此系統的穩定態方程式為

$$C_P = K\, e^{\frac{E}{R}\left[\frac{1000}{910} - \frac{1000}{T+460}\right]} C_A^s\, C_B^t\, \Theta_h$$

其中

C_P = 反應器中產物的濃度 (g mole/liter)

C_A = 反應器中反應物 A 的濃度 (g mole/liter)

C_B = 反應器中反應物 B 的濃度 (g mole/liter)

K = 在 450°F 時的反應速率常數

T = 溫度 (°F)

E = 活化能 /1000

R = 氣體常數

Θ_h = 反應器的滯留時間

s , t = 無因次參數



圖 7.13 連續攪拌反應系統

利用爲實驗目的準備的穩定進料，獲得以下數據；

Run	C_{A0}	C_{B0}	T	θ_h	C_A	C_B	C_P
1	1.0	1.5	400	0.20	0.75	1.19	0.11
2	1.0	1.5	400	0.52	0.67	1.10	0.20
3	1.0	1.5	400	1.00	0.56	1.00	0.30
4	1.0	1.5	450	0.11	0.54	0.96	0.31
5	1.0	1.5	450	0.21	0.45	0.85	0.40
6	1.0	1.5	450	0.38	0.35	0.76	0.50
7	1.0	1.5	450	0.80	0.25	0.68	0.59
8	1.0	1.5	500	0.10	0.24	0.64	0.60
9	1.0	1.5	500	0.21	0.17	0.58	0.65
10	1.0	1.5	500	0.40	0.11	0.54	0.71
11	1.0	0.5	500	0.10	0.65	0.20	0.17
12	1.0	0.5	500	0.30	0.60	0.15	0.23
13	1.0	0.5	500	0.95	0.55	0.09	0.26

(a) 請將穩定態方程式兩邊取對數，證明可以寫成以下型式

$$Y = B_0 + B_1 X_1 + B_2 X_2 + B_3 X_3$$

(b) 利用以上方程式及實驗所得數據估算參數 *K* , *E/R* , *s* 及 t。並求各參數之 97.5%可信任區間。

(c) 若假設反應為 A 的一次反應，B 的二次反應，則穩定態方程式可簡化成

$$C_P = K\, e^{\frac{E}{R}\left[\frac{1000}{910} - \frac{1000}{T+460}\right]} C_A^{1.0}\, C_B^{2.0}\, \Theta_h$$

請將穩定態方程式兩邊取對數，證明可以寫成 $Y = B_0 + B_1 X_1$ 型式，利用此方程式及實驗所得數據估算參數 *K* 及 *E/R*。並求各參數之 97.5%可信任區間。

(d) 比較 (b) 及 (c) 所得結果，並作檢討分析。